Technical Note

Fuzzy Query Language

Robert E. La Quey
San Diego, California

Introduction

The purpose of this technical note is to provide a simple model demonstrating one use of fuzzy sets in FORTH.

Fuzzy sets have become an academic industry. My intent is not to try to review the products of this industry, but to point to their existence and demonstrate that they are easily imported into FORTH. The seminal work is by Zadeh [1978].

Modern mathematics is built up from set theory. Thus, if one redefines the fundamental notion of a set, then mathematics in general needs reconstruction. The proposition has the same profound implications for mathematics as the redefinition of @ and ! has for FORTH.

Zadeh pointed out that classical set theory was rather fascist in its insistence that an element must either belong (1) or not belong (0) to a set. He introduced the idea of a "fuzzy" set which was characterized by allowing elements to have a degree of membership, i.e. a value between 0 and 1, in the set.

A fuzzy set is described by a characteristic function. I chose arbitrarily to let this value vary from 0 to 100. An element that is certainly in the set has a characteristic value 100. An element that is certainly not in the set has a value 0. In between, the value is between 0 and 100. Scaling serves to map the real world into this range.

An Implementation

Screen 9 provides a few words for dealing with this scaling. The fuzzy set itself is represented by a table which allows one to determine the degree of membership given any element. Screen 10 implements this table. Screen 11 associates a set of conventional flags with ranges of values in the table.

Much of this is arbitrary and should be designed to fit a specific problem. For many problems, two or three levels between the limits of `ALWAYS CERTAIN` and `NEVER` are probably adequate. On the other hand artists probably discriminate a much wider range of REDs than do most accountants.

Entire books have been written on the first few lines of Screen 12. The interesting fact seems to be that there are many different ways of choosing `AND OR NOT` which preserve De Morgan's Theorem. (See Charles Moore's digital simulator for instance [MOO84].) The choice presented here is perhaps the simplest and is rather intuitive.

Note that this text is sprinkled with words like much, many, few, probably, most, rather, etc. These are the words that fuzzy sets are aimed at. In Screen 13 we provide a word `.F` which attempts to capture some of this flavor.

A toy application has been included to give the reader a feel for fuzzy sets. One has:

Application = Database + Fuzzy Sets + Query Language.

Screens 18 to 20 lay down a simple database in RAM. Screen 21 defines some fuzzy sets and illustrates how easily they may be combined.

Screens 22 through 25 define a Query Language. The syntax is more like FORTH than English, but the semantics is like English. If you want English-like syntax, then use a syntactical parser [MOR85] on the input stream. One might consider this use of fuzzy sets as an example of how to implement one aspect of a semantic parser (PAR85).

Finally Screens 26 and 27 contain examples of queries that are supported by the fuzzy sets and the use of the query language.

Bibliography

1. Zadeh, L. A. "Fuzzy Sets as a Basis for a Theory of Possibility," *Fuzzy Sets and Systems*, **1**:3-28, 1978.

2. Moore, Charles. "A Language for Digital Design," paper presented at the 1984 FORML Conference, Asilomar, California, November 23-25, 1984.

3. Morgenstern, Leonard. "BNF: A Parser Written in Forth," paper presented at the 1985 FORML Conference, Asilomar, California, November 29-December 1, 1985.

4. Park, Jack. "An Approach to Natural Language Parsing," paper presented at the 1985 FORML Conference, Asilomar, California, November 29-December 1, 1985.

In memoriam: Michael Reagan 1961-1985

Few are wholly dead
Blow on a dead man's embers
And a live flame will start.

Robert Graves

Mike introduced me to fuzzy logic. He was a good FORTH programmer, and he was human.

Manuscript received June 1986.

Appendix

The following source code is written in a variant of Laxen-Perry F83. It is placed in the public domain and may be modified as needed. I would appreciate hearing about interesting improvements users may implement.

```
Scr # 1
  0 ( load screen )
  1
  2   2 LOAD  ( FORTH Extensions     )
  3   9 LOAD  ( Fuzzy set Primitives )
  4  18 LOAD  ( Application = Database + Fuzzy Sets + Query Lang)
  5  26 LOAD  ( Queries )
  6
  7
  8
  9
 10
 11
 12
 13
 14
 15

Scr # 2
  0 ( FORTH EXTEN   6/22/85 rel )
  1
  2  FORTH DEFINITIONS
  3
  4  : KQ KEY 27 = IF QUIT THEN ;
  5  : PRINTER [ HIDDEN ] PR-START ; FORTH
  6  : CONSOLE [ HIDDEN ] PR-STOP  ; FORTH
  7  \ HIDDEN is a vocabulary - for some implementations, PRINTER may
  8  be defined as PRINTING ON.
  9  VARIABLE REPEAT?
 10  : Begin 1 REPEAT? ! >IN @ ;
 11
 12  : Until ( >IN,Flag --- >IN )
 13       IF DUP >IN ! ELSE DROP 0 REPEAT? ! THEN ;     EXIT
 14
 15  Latter constructs allow interpretive looping

Scr # 9
  0 ( Fuzzy Sets    6/22/85 rel )
  1
  2  VARIABLE SET^
  3
  4  : MINX    SET^  @ 2+  @;                ( --- Minvalue      )
  5  : MAXX    SET^  @     @;                ( --- Maxvalue      )
  6  : DX      MAXX  MINX - ;                ( --- Delta val     )
  7  : POINT   SET^  @ 4 +  ;                ( --- Addr Deg Table)
  8  : SCALE   10  DX */ 2* ;                ( Value --- Index   )
  9  : MINDEG  POINT       @;                ( --- Deg at Minval )
 10  : MAXDEG  POINT 20 + @ ;                ( --- Deg at Maxval )
 11  : DEGREE  MINX -   SCALE  POINT + @ ;    ( Value --- Degree  )
 12    -->
 13   These words provide support for computing the Degree of
 14   Membership in a fuzzy set associated with the value of an
 15   element, i.e. the characteristic function of the fuzzy set.
```

```
Scr # 10
  0 ( Fuzzy Sets     6/22/85 rel )
  1
  2  : FUZZY.SET    ( Table, Minvalue, Maxvalue --- )
  3      CREATE 13 0 DO   , LOOP
  4      DOES> SET^ !             ( Value --- Degree of Membership)
  5            DUP MINX <
  6            IF DROP MINDEG
  7            ELSE DUP MAXX >
  8                 IF  DROP MAXDEG ELSE DEGREE THEN
  9            THEN ;                                    -->
 10
 11  At compile time a Table containing 11 elements representing the
 12  degrees of membership in the fuzzy set is placed on the stack
 13  followed by the minimum and maximum values of the input value
 14  of a candidate for membership.  At run time a value is input on
 15  the stack and a degree of membership is returned.

Scr # 11
  0 ( Fuzzy Sets     6/22/85 rel )
  1
  2  : ALWAYS?          98 100 BETWEEN ;      ( --- Flag )
  3  : ALMOST.ALWAYS?   85  97 BETWEEN ;      ( --- Flag )
  4  : VERY.OFTEN?      69  84 BETWEEN ;      ( --- Flag )
  5  : OFTEN?           60  68 BETWEEN ;      ( --- Flag )
  6  : UNSPECIFIC+?     50  59 BETWEEN ;      ( --- Flag )
  7  : UNSPECIFIC-?     40  49 BETWEEN ;      ( --- Flag )
  8  : SELDOM?          31  39 BETWEEN ;      ( --- Flag )
  9  : VERY.SELDOM?     15  30 BETWEEN ;      ( --- Flag )
 10  : ALMOST.NEVER?     2  14 BETWEEN ;      ( --- Flag )
 11  : NEVER?           0   1 BETWEEN ;      ( --- Flag )
 12
 13   -->
 14   Membership tests.  If Degree of Membership is between 85 and 97
 15   then we say something is " almost always " true.

Scr # 12
  0 ( Fuzzy Sets    6/22/85 rel )
  1
  2   100  CONSTANT CERTAIN
  3
  4  : Or  MAX ;     ( Think about it!  Many other choices are     )
  5  : And MIN ;     ( possible and may be desirable on occasion.  )
  6  : Not CERTAIN SWAP - ;
  7
  8  ( Some useful tables )
  9  : SAME          100 90 80 70 60 50 40 30 20 10   0 ;
 10  : MORE          100 81 64 49 36 25 16  9  4  1   0 ;
 11  : LESS          100 95 89 85 78 71 65 55 45 32   0 ;
 12
 13  : INCREASING    100 92 77 65 55 45 36 23  8  2   0 ;
 14  : DECREASING      0  1  8 23 36 45 55 65 77 92 100 ;
 15  : LOCAL           0  2 16 46 72 98 98 72 46 16   0 ;  -->
```

```
Scr # 13
  0 ( Fuzzy Sets    6/22/85 rel )
  1
  2  : DE DROP EXIT ;
  3
  4  : .F ( Degree ---- )
  5      DUP ALWAYS?          IF ." always "              DE   THEN
  6      DUP ALMOST.ALWAYS?   IF ." almost always "       DE   THEN
  7      DUP VERY.OFTEN?      IF ." very often "          DE   THEN
  8      DUP OFTEN?           IF ." often "               DE   THEN
  9      DUP UNSPECIFIC+?     IF ." occasionally "        DE   THEN
 10      DUP UNSPECIFIC-?     IF ." probably would not "  DE   THEN
 11      DUP SELDOM?          IF ." seldom "              DE   THEN
 12      DUP VERY.SELDOM?     IF ." very seldom "         DE   THEN
 13      DUP ALMOST.NEVER?    IF ." almost never "        DE   THEN
 14      DUP NEVER?           IF ." never "               DE   THEN ;
 15      -->

Scr # 14
  0 ( Fuzzy Sets    6/22/85 rel )
  1  VARIABLE 'FUZZY
  2  : SET_FUZZY ' DUP ' FUZZY ! >BODY SET^ ! ;
  3  : FUZZY 'FUZZY @ EXECUTE ;
  4  : .FUZZY_NAME 'FUZZY @ >NAME .ID ;
  5  : .FUZZY FUZZY DUP 5 .R SPACE .F SPACE .FUZZY_NAME ;
  6  : SHOW
  7      SET_FUZZY
  8      CR ." MAX " MAXX 3 .R 5 SPACES ." MIN " 3 .R
  9      MAXX DX 2/ + MINX DX 2/ -
 10      DO   CR I 3 .R I .FUZZY   LOOP  ;
 11
 12  EXIT
 13  Miscellaneous tools.  SHOW is used to display a fuzzy set.
 14  SHOW <fuzzy.set> <CR>    e.g.   SHOW TALL
 15  A good fuzzy.set editor would be useful.

Scr # 18
  0 ( Database      6/22/85 rel )
  1
  2  VARIABLE DATA^  VARIABLE #RECS  VARIABLE REC#  VARIABLE FIELD#
  3
  4  : DATA
  5     HERE DATA^ !     0 #RECS !
  6  0  BEGIN
  7       BL WORD DROP     DUP 4 MOD 0=
  8       IF 1 #RECS +!  5 ALLOT ELSE HERE NUMBER DROP , THEN    1+
  9       >IN @ 1000 >
 10     UNTIL DROP  ;
 11
 12  -->
 13 DATA simply lays the data down in the dictionary
 14
 15
```

```
Scr # 19
  0 ( Database      6/22/85 rel )
  1
  2  DATA   JOHN      18          72          160
  3         JIM       18          75          210
  4         JACK      19          60          180
  5         TOM       25          66          150
  6         DICK      30          69          175
  7         HARRY     35          68          170
  8         GUY       40          70          170
  9         BOB       45          64          205
 10         RICH      50          58          100
 11         RICK      55          69          180
 12         BILL      60          71          150
 13         PHIL      65          66          150
 14         AL        70          68          155
 15         MIKE      71          75          180      -->

Scr # 20
  0 ( Database      6/22/85 rel )
  1
  2  : NAME^ REC# @ 11 * DATA^ @ + ;
  3
  4  : NAME   0 FIELD# ! NAME^       ;
  5  : AGE    1 FIELD# ! NAME^ 5 + @ ;
  6  : HEIGHT 2 FIELD# ! NAME^ 7 + @ ;
  7  : WEIGHT 3 FIELD# ! NAME^ 9 + @ ;          -->
  8
  9    Field definitions
 10
 11
 12
 13
 14
 15

Scr # 21
  0 ( Fuzzy Sets    6/22/85 rel )
  1  INCREASING  64 74 FUZZY.SET TALL
  2  DECREASING  60 66 FUZZY.SET SHORT
  3  LOCAL       64 70 FUZZY.SET AVG.HT
  4  INCREASING  50 65 FUZZY.SET OLD
  5  DECREASING  18 30 FUZZY.SET YOUNG
  6  MORE         0 100 FUZZY.SET VERY
  7  DECREASING  0   0 FUZZY.SET SMALL
  8
  9  : Near 2DUP < IF SWAP THEN 2DUP + >R - 100 R> */ SMALL ;
 10  : MIDDLE     DUP OLD Not SWAP YOUNG Not And ;
 11  : OLDTALL    HEIGHT TALL  AGE OLD   And ;
 12  : YOUNGSHORT HEIGHT SHORT AGE YOUNG And ;
 13  -->
 14   Fuzzy set definitions.  People may differ on the degrees of
 15   membership.
```

```
Scr # 22
  0 ( Query Lang    6/22/85 rel )
  1
  2  VOCABULARY PRONOUN
  3
  4  PRONOUN DEFINITIONS
  5
  6  : HE ;  : HIS ;   : SHE ;   : HER ;
  7  : ANYONE 0 REC# ! FORTH Begin ;
  8
  9   -->
 10
 11
 12
 13
 14
 15

Scr # 23
  0 ( Query Lang    6/22/85 rel )
  1  FORTH DEFINITIONS
  2  : >LINE >IN @ C/L / C/L * ;
  3  : .Q BLK @ NOT
  4       IF CR  BLK @ BLOCK >LINE + C/L 10 - CR TYPE THEN ;
  5
  6  : SEARCH ( stringaddr --- Flag    Searches database )
  7       0 REC# <
  8       BEGIN
  9          REC# @ #RECS @ <
 10       WHILE
 11          DUP NAME^ COUNT COMPARE 0=
 12          IF  DROP TRUE EXIT   ELSE   1 REC# +!   THEN
 13       REPEAT DROP FALSE ;
 14  -->
 15     .Q displays query when LOADing

Scr # 24
  0 ( Query Lang    6/22/5 rel )
  1
  2  : Is .Q
  3       ONLY PRONOUN DEFINED
  4       IF   EXECUTE        ( is a pronoun so execute it      )
  5       ELSE 1+ SEARCH 0=   ( Not a pronoun so search database)
  6            IF CR ." I do not know " HERE COUNT TYPE CR
  7               ASCII ? WORD DROP     ( Skip over rest of query)
  8               ONLY FORTH EXIT       ( Continue interpreting  )
  9            THEN
 10       THEN ONLY FORTH ;
 11  -->
 12  Is parses a word.  If it is in the PRONOUN VOCABULARY then
 13  it is executed.  If not then Is SEARCHs the database to see
 14  if it is a name that it recognizes.
 15
```

```
Scr # 25
  0 ( Query Lang    6/22/85 rel )
  1
  2  : Tell
  3      Is CR NAME^ COUNT TYPE ."  is " AGE . ."  years old. "
  4         ." He is " HEIGHT . ."  inches tall and weighs "
  5         WEIGHT . ."  pounds." CR CR ;
  6  : About Tell 28 LOAD ;
  7  : ? ( Degree ---- )
  8      REPEAT? @
  9      IF    85 > IF CR NAME COUNT TYPE THEN
 10            1 REC# +! REC# @ #RECS @  < Until
 11      ELSE  CR 5 SPACES ." People "  .F ." say so." CR KQ THEN ;
 12  EXIT
 13  Tell is used in the sense of Tell about. e.g. Tell JOHN <CR>
 14  About is Tell followed by the batch of queries on Screen 28.
 15  Change About if you enter queries on other screens.

Scr # 26
  0 ( Queries      6/22/85 rel )
  1 CR
  2 Tell BILL
  3 Is BILL HEIGHT TALL ?
  4 Is HIS  HEIGHT SHORT ?
  5 Is BILL HEIGHT AVG.HT Not ?
  6 Is HER  HEIGHT TALL VERY Not ?
  7 Is BILL HEIGHT TALL Not VERY ?
  8 Is HIS  HEIGHT TALL HEIGHT SHORT Or ?
  9 Is HIS  AGE OLD ?
 10 Is BILL AGE YOUNG ?
 11 Is BILL AGE MIDDLE ?
 12 Is BILL AGE OLD Not HEIGHT TALL And ?
 13 Is ANYONE HEIGHT TALL ? KQ
 14 About DICK  About JACK
 15 About MIKE                                          QUIT

Scr # 28
  0 ( Queries About 6/22/85 rel )
  1 Is HIS  HEIGHT TALL ?
  2 Is HIS  HEIGHT 70 Near ?
  3 Is HIS  HEIGHT SHORT ?
  4 Is HIS  HEIGHT AVG.HT   ?
  5 Is HIS  HEIGHT TALL VERY ?
  6 Is HIS  HEIGHT TALL Not VERY ?
  7 Is HIS  HEIGHT TALL AGE OLD Or ?
  8 Is HIS  AGE OLD ?
  9 Is HIS  AGE YOUNG ?
 10 Is HIS  AGE MIDDLE ?
 11 Is HIS  AGE OLD Not HEIGHT SHORT And ?
 12 Is HIS  AGE YOUNG HEIGHT TALL And ?
 13 Is HE   YOUNGSHORT ?
 14 Is HE   OLDTALL ?
 15 EXIT
```